# Surface area of a cylinder [closed]

Evaluating

expr = {Sin[t], Cos[t], Sin[p]};
ParametricPlot3D[expr, {t, 0, 2 π}, {p, 0, 2 π}]
Area[expr, {t, 0, 2 π}, {p, 0, 2 π}]


gives an area of 8 π.

My question is how to interpret that area? This seems odd to me as the area is double what it should be.

## closed as off-topic by Yves Klett, bbgodfrey, Karsten 7., bobthechemist, Sjoerd C. de VriesMar 3 '15 at 21:16

• The question does not concern the technical computing software Mathematica by Wolfram Research. Please see the help center to find out about the topics that can be asked here.
If this question can be reworded to fit the rules in the help center, please edit the question.

• Yes, let's say they are the correct limits. – user11946 Mar 3 '15 at 7:48
• I am not sure how to see the area from Plot[Sin[p],{p,0,2 Pi}] – user11946 Mar 3 '15 at 8:12
• From the ParametricPlot3D, the circumference is 2 Pi and the height is 2. So I calculated the area of the cylinder as (2 Pi)*2 = 4 Pi. Am I wrong in my approach? Thanks. – user11946 Mar 3 '15 at 8:14
• I'm voting to close this question as off-topic because the alleged problem stems from a misunderstanding of the chosen parametrization. – Yves Klett Mar 3 '15 at 12:57

Let mi visualize this mistake. Instead of constant radius we will use radius that depends of p:

expr = {1 + .1 p, 1 + .1 p, 1} { Sin[t], Cos[t], Sin[p]};

ParametricPlot3D[expr, {t, 0, 3 Pi/2}, {p, 0, 2 \[Pi]}]


As you can see, for p limits: {0, 2Pi} you are plotting your surface twice. Try with Cos[p] and p in {0,Pi} or whatever monotonic function for given interval.

This is documented behavior. The parametric forms of ArcLength, Area, and Volume take the parametrization as fundamental, and compute the area including multiple coverings. The region forms of the functions take the image as fundamental, and compute the area of the embedding into R^n. If you want the latter, you should use ParametricRegion.

From the reference page:

The parametric form of Area computes the area of possibly multiple coverings:

Area[{Cos[φ] Sin[θ], Sin[φ] Sin[θ], Cos[θ]}, {θ, 0, π}, {φ, 0, 4  π}]

8 π


The region version computes the area of the image:

Area[ParametricRegion[{Cos[φ] Sin[θ], Sin[φ] Sin[θ], Cos[θ]}, {{θ, 0, π}, {φ, 0, 4 π}}]]

4 π

Area[Sphere[{0, 0, 0}, 1]]

4 π


I think you should step back and think carefully about the expression you're using. Why are you using Sin[p] to represent the height? Also, if the height is 2, then it is not 2π.

expr = {Sin[t], Cos[t], p};
ParametricPlot3D[
expr,
{t, 0, 2 \[Pi]}, {p, 0, 2},
AxesLabel -> (Style[#, 18, "Label", Blue] & /@ {"x", "y", "p"})
]
Area[expr, {t, 0, 2 \[Pi]}, {p, 0, 2 }]